Graph partitioning: AMPL model (1/2)

```
param n >= 1, integer;  # number of vertices
set V := 1..n;
set E within {V,V};

# edge weights
param c{E};  # edge weights

# vertex colours
param lambda{V};
param gamma{u in V, v in V : u != v} :=
  if (lambda[u] = lambda[v]) then 0 else 1;

# max number of clusters
param kmax;
set K := 1..kmax;


# original problem variables
var x{V,K} binary;

# linearization variables
var w{V,K,V,K} >= 0, <= 1;

# cluster existence variables
var z{K} binary;
```

Graph partitioning: AMPL mod (2/2)

```
minimize intercluster :
  sum{k in K, l in K, (u,v) in E : k != l} c[u,v] * w[u,k,v,l] +
  sum{k in K} z[k];

# constraints
subject to assignment {v in V} : sum{k in K} x[v,k] = 1;

subject to cardinality {k in K} : sum{v in V} x[v,k] <= ceil(card{V}/2);

subject to existence {k in K} : sum{v in V} x[v,k] >= z[k];

subject to diffcolours {u in V, v in V, k in K, l in K : u != v and k != l} :
  w[u,k,v,l] <= gamma[u,v];

# linearization constraints

subject to lin1 {u in V, v in V, h in K, k in K : (u,v) in E or (v,u) in E} :
  w[u,h,v,k] <= x[u,h];

subject to lin2 {u in V, v in V, h in K, k in K : (u,v) in E or (v,u) in E} :
  w[u,h,v,k] <= x[v,k];

subject to lin3 {u in V, v in V, h in K, k in K : (u,v) in E or (v,u) in E} :
  w[u,h,v,k] >= x[u,h] + x[v,k] - 1;
```

Graph partitioning: AMPL dat

```
param n := 16; # number of vertices

param kmax := 4;  # max number of clusters

param : E :   c :=
       1 15 1
       2 15 1
       2  3 1
       2  4 1
       3  5 1
       4  5 1
       5  6 1
       5 16 1
       6  9 1
       7  8 1
       7 16 1
       8 10 1
       8 16 1
       11 16 1
       12 16 1
       13 16 1
       14 16 1
;
```

Graph partitioning: AMPL dat

```
param lambda :=
1 1
2 2
3 3
4 3
5 2
6 2
7 2
8 2
9 1
10 1
11 4
12 4
13 4
14 4
15 1
16 4
;
```

Graph partitioning: AMPL run

```
model clustering.mod;
data random.dat;

# solver:
option solver cplex;

# solving the problem
solve;

# printing the result
display x;
```

Graph partitioning: Solution

```
ILOG AMPL 11.010, licensed to "ecolepolytechnique-palaiseau".
AMPL Version 20080219 (Linux 2.6.9-5.ELsmp)
ILOG CPLEX 10.100, licensed to "ecolepolytechnique-palaiseau", options: e m b q use=
CPLEX 10.1.0: optimal integer solution; objective 1
65 MIP simplex iterations
0 branch-and-bound nodes
x [*,*]
:    1   2   3   4    :=
1    1   0   0   0
2    1   0   0   0
3    1   0   0   0
4    1   0   0   0
5    1   0   0   0
6    1   0   0   0
7    0   1   0   0
8    0   1   0   0
9    1   0   0   0
10   0   1   0   0
11   0   1   0   0
12   0   1   0   0
13   0   1   0   0
14   0   1   0   0
15   1   0   0   0
16   0   1   0   0
;
```